 A product of integer random variables

Sooner or later when studying processes that contain a random component there is a need to study functions of random variables. Sometimes the function is a simple sum, sometimes the function is more complicated. To derive exactly the expected value, the standard deviation and the distribution of the function is sometimes an easy task (as for simple sums of variables) or more difficult when the function is of more complicate nature.
There are of course more or less advanced mathematical tools in order to derive the wan​ted result. There are also some approximations that can be used and even simulation is a tool to consider. 

Here we will look at a special, simplified, example where we have a product of two integer-valued variables. In the document Random sums.doc about random sums of variables we have more examples.

The table below contains four different columns representing all possible outcomes. The first column X contains the five pos​sible outcomes from a binomial distribution where n = 4. This distribution gives us the num​ber of e.g. incorrect items in samples of 4. Of course the smallest number is then 0 and the highest number is 4. The second column Y contains the same outcome but for a binomial distribution when n = 3. 

The binomial distribution is well described in the documents A course in statistics, A collection of variables and Exercises with computer support. Also, the macros %Bpdfcdf, %Bhist, %Bdata, %SimB and %MultDist describes the binomial distribution in various ways.

The third column of the table contains the product XY and the fourth column contains the proba​bility for each row calculated as a simple product. Here e.g. P(X = 0) means the probability that the X-variable obtains the value 0 and P(Y = 2) means the probability that the Y-variable obtains the value 2. (If n = 3 and p = 0.2 we would read it in plain language as “the probability that we find exactly 2 faulty items in a batch of 3 items when the fault rate is 0.2.”)

The probability distribution of the discrete binomial distribution is calculated using the following expression:

	where X is the variable, x is its numerical value, p is the fault rate (0 < p < 1).

Ex.: 5! = 5·4·3·2·1 = 120

Next step is to sum the probability of P(XY = 0), P(XY = 1),…, P(XY = 12). Observe that not all integers from 0 to 12 are possible. E.g. the values 7 and 10 are not possible. After doing this, we have the probability distribution for the product of the two binomially distributed variables.

The probability distribution.  Below we use Minitab to calculate the probability distribu​tion of XY. Then we can calculate the expected value  and standard deviation .

name c1 'X' c2 'P(X)'         # Names the columns used.

name c4 'Y' c5 'P(Y)'

name c7 'XY' c8 'P(XY)'

name c10 'Z' c11 'P(Z)'       # NB: Z = XY.

set c1                        # Fills c1 with 0, 0, 0, 0 to

(0:4)4                        # 4, 4, 4, 4, according to the

end                           # table above.

pdf c1 c2;                    # Stores the probabilities of the

binomial 4 0.1.               # X-values in c2 (Bin(4, 0.1).

set c4                        # Fills c4 with 0, 1, 2, 3 according

5(0:3)                        # Y-values above.

end

pdf c4 c5;                    # Stores the probabilities of the

binomial 3 0.2.               # Y-values in c5 (Bin(3, 0.2).

let c7 = c1*c4                # Calculates the XY-values.

let c8 = c2*c5                # Multiplies the probabilities.

stats c7;                     # A way to create the different

by c7;                        # possible XY-values in c10.

mean c10.

stats c8;                     # Sums the corresponding proba-

by c7;                        # bilities. The result is stored

sums c11.                     # c11.
The data window now contains 8 columns and column c10 and c11 contains the probability distri​bution of the product. The other columns are of no further value.

Plotting the distribution.  The next natural step is to plot the resulting discrete distribution. We use some of the available commands to design the plot the way we want it: 

plot c11*c10;                       # Plots the prob aganst the Z.

title 'Probability distribution of Z = X*Y'; # Title of the plot.

tcolor 4;                           # Gives blue text of title.

tsize 1.1;                          # The size of the text.

title 'X is Bin[4; 0.1] and Y is Bin[3; 0.2]'; # 2nd row of title.

tcolor 2;                           # Gives red text of title.

tsize 0.9;                          # The3 size of the text.

project;                            # Projects a line downwards.

base 0;                             # Line goes down to zero level.

size 10;                            # With of vertical line (bar).

color 2;                            # Gives red bars.

data 0.15 0.90 0.15 0.85;           # The size of data window.

etype 0;                            # Removes frame around data window.

axlabel 1 'Values of the variable Z'; #

tsize 0.7;                          # text size 0.7.

axlabel 2 'Probability';            # 

tsize 0.7;                          # text size 0.7.

Scale 1;

tsize 0.6;

length 0.005;

HDisplay 0 0 0 0;

Scale 2;
min 0;
length 0.005;

tsize 0.6;

HDisplay 0 0 0 0;

nodt.
The graph shows a distribution that is dominated by the first three values (0, 1, 2) and some are missing completely (e.g. 5 is not a possible number) and some are too small to be visible on the graph.

Calculating mu and sigma.  Now we know the probabilities and thus we can calculate the expected value () and the standard deviation (). The following formulas, that are valid for any discrete variable, are used (see e.g. chapter 7 in the Ing-Stat document A course in statistics):

In order to do verify the calculations we use the %CreDist-macro that also simulates data using the input data. Use the following commands in the session-window of Minitab after creating the distri​bution according to the instructions above:

· Start another session of Minitab

· Run the macro %CreDist once

· Copy and paste the contents of column c10 and c11
from the original worksheet i.e. Z and P(Z) into c1 and c2 of the new session

· State the number of values to be simulated (last row of c1)

· Rerun the macro %CreDist with the new c1 and c2

The graph 'Simulating a discrete distribution (fig 2)' shows the information. The theoretical distribution and the simulated result coincide probably. See also the information of the graph 'Simulating a discrete distribution (fig 1)' and '(fig 3)'.

The result gives that  = 0.24 and  = 0.616. Below we will simulate the situation in a slightly different way.

Simulation.  The simulation above via the %CreDist-macro is performed using the input via column c10 and c11 calculated above. We should of course also simulate the Z-variable via the original situa​tion, i.e. as the product of two binomial variables. 

We do this via the following commands:

restart                       # Restarts the Minitab session.

random 5000 c12;              # Stores in c12 5000 values from

binomial 4 0.1.               # Bin(4, 0.1).

random 5000 c13;              # Stores in c13 5000 values from

binomial 3 0.2.               # Bin(3, 0.2).

let c14 = c12*c13             # Calculates the observed XY-values.

hist c14;                     # Creates a histogram of the XY-data.

density.                      # as a distribution (i.e. the sum of 

                              # the bars is 1).

describe c14                  # Describes the data in XY-data.

A visual comparison of the plot of the distribution and the histogram shows good agreement. If we want to do the comparison numerically we can use the following commands:

tally c14;                    # Makes a table of the results stored

store c15 c16.                # in c15 and c16.

let c17 = c16/sum(c16)        # Calculates the observed density.

The result in c17 is probably very close to numerical result in c11 (the theoretical distribution).

A larger example.  We also try the theory and the reasoning above using a larger example. We use now a binomial distribution where n = 100 and p = 0.1 for the illustrations. (NB n is stored in the constant k1 below. If the software has insufficient memory available, one possibility is to decrease n by changing the value of the constant k1.)

restart                       # Restarts the Minitab session.

name c1 'X' c2 'P(X)'         # Names the columns used.

name c4 'Y' c5 'P(Y)'

name c7 'XY' c8 'P(XY)'

name c10 'Z' c11 'P(Z)'       # NB: Z = XY.

let k1 = 100                  # 'n' in the binomial distribution.

let k2 = k1 + 1               # -

set c1                        # Fills c1 with 0, 0, 0, 0 to

(0:k1)k2                      # 100, 100, 100.

end                           # -

pdf c1 c2;                    # Stores the probabilities of the

binomial k1 0.1.              # X-values in c2 (Bin(100, 0.1).

set c4                        # Fills c3 with 0, 1, 2, , 100.

k2(0:k1)                      # -

end

pdf c4 c5;                    # Stores the probabilities of the

binomial k1 0.1.              # Y-values in c4 (Bin(100, 0.1).

let c7 = c1*c4                # Calculates the XY-values.

let c8 = c2*c5                # Multiplies the probabilities.

stats c7;                     # A way to create the different

by c7;                        # possible XY-values in c7.

mean c10.

stats c8;                     # Sums the corresponding proba-

by c7;                        # bilities. The result is stored

sums c11.                     # c11.

In order to do the plotting of the distribution of the larger example we first get rid of all probabi​lities less than, say, 0.0001. Otherwise we will get a distribution that squeezed to the left edge of the diagram:

copy c11 c10 c11 c10;         # Copies c11 and c10 back to c11 c10

omit c11 0:0.0001.            # omitting the range 0 to 0.0001.

plot c11*c10;                 # Plots the prob aganst the Z.

title 'Probability distribution of Z = X*Y'; # Title of the plot.

tcolor 4;                     # Gives blue text of title.

tsize 1.1;                    # The size of the text.

title 'X is Bin[k1; 0.1] and Y is Bin[k1; 0.1]'; # 2nd row of title.

tcolor 2;                     # Gives red text of title.

tsize 0.9;                    # The3 size of the text.

project;                      # Projects a line downwards.

base 0;                       # Line goes down to zero level.

size 4;                       # With of vertical line (bar).

color 2;                      # Gives red bars.

data 0.15 0.90 0.15 0.85;     # The size of data window.

etype 0;                      # Removes frame around data window.

axlabel 1 'Values of the variable Z'; # 

tsize 0.7;                    # text size 0.7.

axlabel 2 'Probability';      #

tsize 0.7;                    # text size 0.7.

Scale 1;

tsize 0.6;

length 0.005;

HDisplay 0 0 0 0;

Scale 2;

length 0.005;

tsize 0.6;

HDisplay 0 0 0 0;

nodt.
Simulation of the distribution of the larger example:

random 5000 c12;              # Stores in c12 5000 values from

binomial k1 0.1.              # Bin(100, 0.1).

random 5000 c13;              # Stores in c13 5000 values from

binomial k1 0.1.              # Bin(100, 0.1).

let c14 = c12*c13             # Calculates the observed XY-values.

hist c14;                     # Creates a histogram of the XY-data.

density.

describe c14                  # Describes the data in XY-data.

Again there is a good agreement of the theoretical and the simulated result.

Final remark.  This exercise shows how it is possible, at least sometimes, to derive e.g. ex​pected value, standard deviation and distribution in a fairly complicated situation. We have to admit that even if the progress above builds on simple steps we rely heavily on the computer in order to get the results.  ■

